

HOW TO... Plot OS grid references on a map

Mike Bedford explains how a few simple Excel formulae can help you to plot Ordnance Survey grid references on a map

Just been on a walk with a GPS receiver and want to see where you've been? Or maybe you're planning a day's hiking and you've downloaded a series of waypoints from a website and want to see the route on a map? Dedicated mapping software can do this for you, but it can be expensive. Happily, you can achieve the same thing with Excel.

The idea is simple enough. Create an Excel workbook with the eastings in one column and the northings in another and use the Chart Wizard to plot them as an XY plot using a map as the background. In practice things are more involved because an Ordnance Survey grid reference isn't just a pair of numbers representing the easting and the northing. Instead, it's a combination of letters and numbers (such as SE001943) that must first be converted into numbers. Here, we show you how to work with the common six-figure grid references

like this one. We'll plot grid references as isolated points on a map, but if you want to show a route the only difference is in Step 7, where you'd select a different chart type.

1 Find the files OSGridRef.xls and OSmap.tif on this month's cover CD and copy them to your hard disk. The Excel file contains some sample grid references for us to work with, which correspond to the locations of caves in the Yorkshire Dales. The area these grid references refer to is covered by our excerpt from the Ordnance Survey map. You can just as easily use your own grid references and plot these on the OS map of the relevant area. To do so, simply replace our example grid references with your own and scan your OS map with a flatbed scanner. Trim the scanned image so its edges are on the map's gridlines, as this will help later on.

PLOTTING GRID REFERENCES AT A GLANCE

Grid references always begin with two letters but may have four, six, eight or even 10 digits, depending on how precise the location is. A four-digit number (more accurately, a pair of two-digit numbers) is accurate to the nearest kilometre, while a 10-digit number is accurate to the nearest metre.

To adapt this workbook for different grid reference lengths, you'll need to change the formulae in columns E and H to pick the right digits; for an eight-digit grid reference, for example, you'll need to enter `=VALUE(MID(B4,3,4))` into E4 and `=VALUE(RIGHT(B4,4))` into H4. You'll also need to divide the resulting values by 100 rather than 10 when calculating the eastings and northings.

2 To convert the grid references into eastings and northings, first you need to decipher the first two alphabet characters. The first letter (which can be A to Z excluding I) represents a 500x500km square in the UK; the second letter is a 100x100km square within the 500km square. There's no simple formula to convert these letters into numbers, so we've provided the necessary lookup table on a separate worksheet in the Excel file. Click on the AtoZ tab to see it. The table includes each possible letter together with the offset it represents as an easting and a northing.

3 Click on the Main tab to return to the main worksheet, select cell C4 and enter the following formula (with no spaces):

```
=LOOKUP(LEFT(B4,1),AtoZ!$A$4:$A$28,
AtoZ!$B$4:$B$28)
```

This tells Excel to examine the left-most character in cell B4, find the matching value in cells A4 to A28 in the AtoZ worksheet and use the corresponding value in B4 to B28. The dollar signs are used so that, when we copy the formula down for the other grid references, the references to the cells on sheet AtoZ remain fixed. The reference to cell B4, on the other hand, will change to B5, B6 and so on. To find a value for the second character, enter the following formula in cell D4:

```
=LOOKUP(MID(B4,2,1),AtoZ!$A$4:$A$28,
AtoZ!$B$4:$B$28)
```

The only difference between the two formulae is that LEFT(B4,1) has been changed to MID(B4,2,1). This tells Excel to pick out a mid-section of B4, two characters in and one character long – in other words, the second letter of the grid reference.

4 In cell E4 we simply want the first three of the six digits in B4 – in Excel this is expressed as the mid-section of B4, starting three characters in and lasting for three characters. We can get this with the formula:

```
=VALUE(MID(B4,3,3))
```

5 To find the values needed for the northings calculations, we'll use the same formulae, except that we want to pick values from column C rather than column B in the AtoZ worksheet, so enter the following formulae into F4 and G4 respectively:

```
=LOOKUP(LEFT(B4,1),AtoZ!$A$4:$A$28,
AtoZ!$C$4:$C$28)
=LOOKUP(MID(B4,2,1),AtoZ!$A$4:$A$28,
AtoZ!$C$4:$C$28)
```

To save having to type these in you can select cells C4 and D4, press Ctrl-C to copy, select cell F4 and type Ctrl-V to paste. You'll then need to change \$B\$4:\$B\$28 at the end of each to \$C\$4:\$C\$28. You'll also need to re-enter the references to B4 at the beginning, as the act of copying and pasting will have changed these to E4. In cell H4 we want the last three digits in B4, so enter the formula:

```
=VALUE(RIGHT(B4,3))
```

6 We're now ready to work out the eastings and northings. The first letter tells us which 500x500km square our grid reference is in, so to get a figure in kilometres we need to multiply the resulting values (in cells C4 and F4) by 500. The second letter is a 100x100km square within the larger square, so we'll multiply these values (D4 and G4) by 100. The final values are measurements to a tenth of a kilometre within the 100x100km square, so we divide these figures by 10. So to arrive at an easting and northing for our first grid reference, enter these formulae into I4 and J4 respectively:

```
=C4*500+D4*100+E4/10
=F4*500+G4*100+H4/10
```

To repeat the same calculations for the other grid references, click and drag to select from C4 to J4. A small box will appear in the bottom-right of the selection; click and drag this down to row 70 or for however many grid references you have.

7 Select the block of cells containing all the eastings and northings (in our case I4 to J70). Click on the Chart Wizard icon, select XY (Scatter) as the Chart type and choose the appropriate Chart sub-type which, for isolated points, is the default. Click on Next and Next again

and then, in Step 3 of the Chart Wizard, click on the Gridlines and Legend tabs and deselect all the boxes so no gridlines or axis legends are displayed. Click on Finish to plot the points on a grey background.

8 Right-click on the grey chart area, select Format Plot Area... from the pop-up menu and click on Fill Effects... in the Format Plot Area window. Select the Picture tab, click on Select Picture and then browse to OSmap.tif or another map that relates to your chosen grid references. Click on Insert followed by OK and OK again. Your chart will now be displayed on the map but your points won't line up.

9 Work out the easting and northing of the bottom-left and top-right corners of your map by reading off the grid references of these corners – for our map, they're SD710740 and SD770780. Put these into two spare cells at the bottom of column B and copy the formulae into the adjacent cells in columns C to J to find out these locations' eastings and northings.

Right-click on the horizontal line at the bottom of the chart and select Format Axis... from the pop-up menu. Select the Scale tab in the window that appears and enter the easting of the bottom-left corner (1371) into the Minimum box and the easting of the top-right corner (1377) into the Maximum box and click OK. Right-click on the vertical line at the left of the chart and do the same, this time with the northing of the bottom-left (974) and top-right corners respectively (978).

The map might not be shown at its correct aspect ratio so, if necessary, select the chart and stretch it so that the gridlines display as squares rather than rectangular boxes. **CS**


Ordnance Survey mapping © Crown copyright. All rights reserved. Media licence 700/05.

HINTS & TIPS OFFICE

HOW TO... Link data between Excel and Word

There aren't many office-related tasks that can't be handled by either Microsoft Word or Excel, but some jobs, such as preparing reports for a client, are best achieved with a combination of the two programs. When doing so, at some point you'll probably want to copy data from one application to the other. Selecting it, typing Ctrl-C to copy and Ctrl-V to paste works well enough, but if the original data gets updated it can be a pain to have to re-import it repeatedly.

To avoid this, copy in the normal way, then in the other document, click Edit, Paste Special... and click the Paste link: option in the dialog box that appears. Now any changes made to the original data will be mirrored in the document into which the data has been pasted. This also works between two Excel files or two Word files. However, bear in mind that moving or changing the name of the original document will break the link.



HINTS & TIPS WORLD WIDE WEB

HOW TO... Discover geographical treasures

The British countryside is a wonderful thing. There's the blissful Brecon Beacons, the stark majesty of the Yorkshire Moors, and other, less celebrated places, such as Golden Balls near Abingdon, or Dancing Dicks just outside Chelmsford. Thundergry on the picturesque Isle of Arran is another one not to be missed.

If you love cartography just as much as childish double entendres, <http://places.jump-around.com> can appease both appetites. Click on the link called Find places close to you, type in your postcode and you'll be presented with a list of ten locations near your home with suitably dubious names. Croydon doesn't feel so gloomy when you discover it's only 6.6 miles from Cock Pond, or 8.1 miles from Titsey Park. Click on the links marked map to open the relevant page on www.multimap.com if you can't quite believe the site's designers didn't make it up.

